The Basic Ant Colony example

In this lesson we will develop a model where ants find a food source. Though each ant follows a set of simple rules, the colony as a whole acts in a sophisticated way. Before starting have a look at this video: https://youtu.be/vG-QZOTc5_Q

HOW IT WORKS

When an ant finds a piece of food, it carries the food back to the nest, dropping a chemical as it moves. When other ants "sniff" the chemical, they follow the chemical toward the food. As more ants carry food to the nest, they reinforce the chemical trail.

Step 1 – Setup world specification

Go to model settings and create a 71 patch squared world with no wrapping, a patch size of 7 and the origin in the centre.

Step 2 – Add interface elements

Add the following interface elements:

- Button calling a procedure called **setup**
- Button calling a procedure called **go**
- Slider called **population** with a range of 1-200 and an increment of 1
- Slider called **diffusion-rate** a range of 0-99 and an increment of 1
- Slider called **evaporation-rate** a range of 0-99 and an increment of 1
- Plot called **Food** with an x axis labelled as time and y as Quantity the code should read as follows:

Pen update commands

```
plotxy ticks sum [food] of patches with [pcolor = cyan]
```

Step 3 – Setup patch variables

Add the following patch variables to store world parameters:

```
patches-own [
  chemical             ;; amount of chemical on this patch
  food                 ;; amount of food on this patch (0, 1, or 2)
  nest?                ;; true on nest patches, false elsewhere
  nest-scent           ;; number that is higher closer to the nest
]
```

Step 4 – Write setup function

Write the setup function to create ants and call further function:

```
to setup
  clear-all
  set-default-shape turtles "bug"
  create-turtles 100 [
    set size 2
    set color red      ;; red = not carrying food
  ]
  setup-patches
  reset-ticks
end
```

Step 5 – Setup patches

Write the following code to set the parameters of the nest and food sources:

```
to setup-patches
  ask patches [
    ;; setup nest and spread a nest-scent over the whole world -- stronger near the nest
    set nest? (distancexy 0 0) < 5
    set nest-scent 100 - distancexy 0 0

    ;setup food
    setup-food -25 20 5 2
    setup-food 20 20 5 1
    setup-food 20 -30 5 3

    recolor-patch
  ]
end
```

Step 6 – Setup food sources

Write the following code to setup the food sources:

```
to setup-food [x y location_size location_amount]  ;; patch procedure
  if (distancexy x y) < location_size [
    set food location_amount
  ]
end
```

Step 7 – setup color patches for visualisations

With the following code to update the patch colors based on the world parameters and chemicals deposited:

```
to recolor-patch  ;; patch procedure
  ;; give color to nest and food sources
  ifelse nest? [
    set pcolor violet
  ][
    ifelse food > 0 [
      set pcolor cyan
    ][
      ;; scale color to show chemical concentration
      set pcolor scale-color green chemical 0.1 5
    ]
  ]
end
```

Step 8 – Test your model

It should look like this:



Step 9 – Create the go function

Write the following code to make your turtles move. 2 critical elements in this model are time and delaying the release of the ants. Time will be used for pheromone dissipation and the delay will increase the efficiency of the ants.

```
to go  ;; forever button
  ask turtles [
    if who >= ticks [
      stop                    ;; delay initial departure
    ]

    ; random wandering
    rt random 40
    lt random 40
    if not can-move? 1 [
      rt 180
    ]


    fd 1
  ]
  tick
end
```

Step 10 – Test your model

Your ants should wander around randomly

Step 11 – Collecting food

Add the following code within ask turtles so your ants can collect food and show this in their appearance:

```
    ; looking for food or nest
    if color = red [
      ;; go in the direction where the chemical smell is strongest
      if food > 0 [
        set color orange + 1      ;; pick up food
        set food food - 1         ;; and reduce the food source
        rt 180                    ;; and turn around
        stop
      ]
    ]
```

Your ants should now collect food, turn orange and turn 180 degrees.

Step 12 – Updating visualisations

Add the following code in the go function but outside of the ask turtles section to update the patch colors to show the removal of food and update the chemical values of the patches:

```
  ask patches [
    set chemical chemical * (100 - evaporation-rate) / 100  ;; slowly evaporate chemical
    recolor-patch
  ]
```

Step 13 – Test your model

The food sources should now appear to be reducing when the ants are picking it up.

Step 14 – searching for scents

Write the following code to facilitate the search for either the nest or food based on the scent being searched for. This code is effectively getting the ant to sniff left and right before moving forward base on what it smells.

```
;; sniff left and right, and go where the strongest smell is
to search-chemical [c_type] ;; turtle procedure
  let scent-ahead scent-at-angle   0 c_type
  let scent-right scent-at-angle  45 c_type
  let scent-left  scent-at-angle -45 c_type

  if (scent-right > scent-ahead) or (scent-left > scent-ahead) [
    ifelse scent-right > scent-left [
      rt 45
    ][
      lt 45
    ]
  ]
end

to-report scent-at-angle [angle c_type]
  let p patch-right-and-ahead angle 1
  if p = nobody [ report 0 ]
  ifelse c_type = "nest" [
    report [nest-scent] of p
  ][
    report [chemical] of p
  ]
end
```

Step 15 – Calling the search chemical function to find food

Add the following code within the if color = red if statement to search for food.

```
      ;; go in the direction where the chemical smell is strongest
      if (chemical >= 0.05) and (chemical < 2) [
        search-chemical "food"
      ]
```

Step 16 – Creating a return to nest function

Write the following function to allow your ant to find their way back to the nest when they have food.

```
to return-to-nest               ;; turtle procedure
  ifelse nest? [
    set color red               ;; drop food and head out again
    rt 180
  ][
    set chemical chemical + 60  ;; drop some chemical
    search-chemical "nest"      ;; head toward the greatest value of nest-scent
  ]
end
```

Step 17 – calling the return to nest function

Change the if color = red to ifelse and the following code to call the newly created return to nest function:

```
    ][
      return-to-nest          ;; carrying food? take it back to nest
    ]
```

Step 18 – Test your model

It should look something like this:



Step 19 – Diffusing the chemical

In the go function add the following line of code to diffuse the chemical:

```
diffuse chemical (diffusion-rate / 100)
```

Step 20 – Play with your model

Experiment with the parameters and see how the sliders impact the model. It should look something like this:



Reference

Base on: Wilensky, U. (1997). NetLogo Ants model. http://ccl.northwestern.edu/netlogo/models/Ants. Center for Connected Learning and Computer-Based Modeling, Northwestern University, Evanston, IL.